9.2.2 Examples to show Bessel's inequality and the completeness relation.

The Fourier coefficients must be computed for the normalized basis functions $\{1/\sqrt{2\,\pi}, \cos(n\pi)/\sqrt{\pi}\}$ for an even function f(x), or $\{\sin(n\pi)/\sqrt{\pi}\}$ for an odd function f(x).

```
Needs["Graphics`Graphics`"]
sqp = Sqrt[π];
sq2p = Sqrt[2 π];
```

■ General Exponent α

```
fa = (π^2 - x^2)^α / π^(2 α);
```

■ Fourier coefficients (Watson 48 (3))

```
ana = Integrate[fa Cos[n x], {x, -π, π},
   Assumptions → Element[n, Integers] && n > 0 && α > - 1] / sqp
```

$$2^{\frac{1}{2}+\alpha}\, n^{-\frac{1}{2}-\alpha}\, \pi^{\frac{1}{2}-\alpha}\, \text{BesselJ}\left[\frac{1}{2}+\alpha, n\,\pi\right] \text{Gamma}[1+\alpha]$$

```
an0 = Integrate[fa , {x, -π, π}, Assumptions →  α > - 1] / sq2p
```

$$\frac{\pi\, \text{Gamma}[1+\alpha]}{\sqrt{2}\, \text{Gamma}\left[\frac{3}{2}+\alpha\right]}$$

■ $\alpha =$ 3/4

```
sa = α → 3 / 4 ;
f = fa /. sa
```

$$\frac{(\pi^2 - x^2)^{3/4}}{\pi^{3/2}}$$

```
pf = Plot[f, {x, -π, π}, AxesLabel → {"x", f}, Ticks → {PiScale, Automatic}];
```



■ Fourier coefficients for normalized basis functions

```
tan = Integrate[f Cos[n x], {x, -π, π}, Assumptions → Element[n, Integers] && n > 0] / sqp
```

$$\frac{2\left(\frac{2}{\pi}\right)^{1/4} \text{BesselJ}\left[\frac{5}{4}, n\,\pi\right] \text{Gamma}\left[\frac{7}{4}\right]}{n^{5/4}}$$

```
fun /. sa
```

$$\frac{2\left(\frac{2}{\pi}\right)^{1/4} \text{BesselJ}\left[\frac{5}{4}, n\,\pi\right] \text{Cos}[n\,x]\, \text{Gamma}\left[\frac{7}{4}\right]}{n^{5/4}}$$

```
a0 = Integrate[f , {x, -π, π}, Assumptions → Element[n, Integers] && n > 0] / sq2p
```

$$\frac{\pi\, \text{Gamma}\left[\frac{7}{4}\right]}{\sqrt{2}\, \text{Gamma}\left[\frac{9}{4}\right]}$$

```
N[%]
```

1.80198

```
ntan = Prepend[Table[tan, {n, 1000}], a0] // N;
```

The first 20 Fourier coefficients are;

```
Take[ntan, 20]
```

```
{1.80198, 0.631629, -0.194824, 0.0971032, -0.0590944, 0.0401561, -0.0292681,
 0.0223927, -0.0177531, 0.0144633, -0.0120393, 0.0101977, -0.00876295, 0.00762174,
 -0.00669784, 0.00593849, -0.00530616, 0.00477355, -0.00432038, 0.00393132}
```

The 1000-th Fourier coefficient is:

```
ntan[[-1]]
```

-3.83962×10^{-6}

■ Check of completeness relation

```
nf = Integrate[f^2, {x, -π, π}]
```

$\frac{3\,\pi^2}{8}$

```
N[%]
```

3.7011

Sum of squared Fourier coefficients taking 1000 terms:

```
Apply[Plus, ntan^2]
```

3.7011

■ Skew-symmetric Function

```
g = D[f, x] 2 / 3
```

$-\frac{x}{\pi^{3/2}\,(\pi^2 - x^2)^{1/4}}$

```
pg = Plot[g, {x, -π, π}, AxesLabel → {"x", g}, Ticks → {PiScale, Automatic}];
```



■ Fourier coefficients for normalized basis functions

```
tbn = Integrate[g Sin[n x], {x, -π, π}, Assumptions → Element[n, Integers] && n > 0] / sqp
```

$-\frac{\left(\frac{2}{\pi}\right)^{1/4}\,\mathrm{BesselJ}\left[\frac{5}{4}, n\,\pi\right]\,\mathrm{Gamma}\left[\frac{3}{4}\right]}{n^{1/4}}$

■ Check of completeness relation

```
ng = Integrate[g^2, {x, -π, π}]
```

$\frac{1}{2}$

```
N[%]
```

```
0.5
```

```
nt = 5000;
ntbn = Table[tbn, {n, nt}] // N;
```

Sum of squared Fourier coefficients taking 1000, 2000, ..., 5000 terms:

```
dnt = 1000;
Table[Apply[Plus, Take[ntbn, nttt // Evaluate] ^2], {nttt, dnt, nt, dnt}]
```

```
{0.486897, 0.490733, 0.492434, 0.493447, 0.494139}
```